MPAE–W–100–94–27

* MAX–PLANCK–INSTITUT FÜR AERONOMIE

D–37189 Katlenburg-Lindau, Federal Republic of Germany

The Tetrahedron Quality Factors of CSDS

PATRICK W. DALY

1994 June 7

This report available from

http://www.mpae.gwdg.de/~daly/tetra.html

Abstract

The four Cluster spacecraft will form a tetrahedron, which ideally should be a regular one: equal spacing between all pairs of vertices. In reality, this will not be the case. A number of parameters exist to specify how badly off the true figure is. This paper presents some mathematics of tetrahedrons, describes the two parameters that are to be used in the Cluster Science Data System, and gives a Fortran program for their calculation.

Introduction Four points in space define a tetrahedron. If the separations between each pair of points are equal, then it is a regular tetrahedron. The four Cluster spacecraft will form a tetrahedron, which in general is not regular. How can we specify the degree to which regularity is achieved?

The “Glassmeier” Parameter The parameter proposed by vom Stein, Glassmeier, and Dunlop (1992) is defined as

Q_G =

True Vol. Ideal Vol. + True Surf. Ideal Surf.

+ 1 (1)

and takes on values between 1 and 3. It tends to describe the dimensionality of the figure, as listed in Table 1. The ideal volume and surface are calculated for a regular tetrahedron with a side length equal to the average of the 6 distances between the 4 points.

Table 1. Special values of the Glassmeier parame- ter Q_G Meaning 1.0 The four points are colinear 2.0 The points all lie in a plane 3.0 A regular tetrahedron is formed

The “Robert/Roux” Parameter

In their paper on tetrahedron shape, Robert and Roux (1993) present 17 different parameters, as ratios of various volumes, sizes, areas. Of these, the CSDS community has decided to adopt one as its second quality parameter for the auxiliary data. It is defined as

Q_R = N · ` True Vol. Sphere Vol.

´ 1_3 (2) where the sphere is that circumscribing the tetrahedron (all four points on its surface) and N is a 1 normalization factor to make Q_R = 1 for a regular tetrahedron. The range of values is between 0 and 1. Mathematics of a Tetrahedron Consider four points in space and the figure formed by joining them with lines (Figure 1). The points are numbered 0 to 3, and their vectors are r_0, r_1, r_2, r_3. Without any loss of generality, we may consider only the differences

d_i = r_i − r_0

in describing the points.

Area of a Side

The area of a parallelogram bounded by two vectors d_1 and d_2 is given by the magnitude of their cross product; any triangle is half of a parallelogram, so its area is S = 1 2 |d_1 × d_2|

where d_1 and d_2 are the vectors for any two sides of the triangle (Figure 2).

For the four sides of the tetrahedron, specify side n to be that one that does not contain point n at any of its vertices. Thus:

S_1 = 1 2 |d_2 × d_3| (3)

S_2 = 1 2 |d_1 × d_3| (4)

0 1 2

3

Figure 1. A tetrahedron and its four vertices.

d_1 d_2 S

Figure 2. Area S of a triangle.

S_3 = 1 2 |d_1 × d_2| (5)

S_0 = 1 2 |(d_2 − d_1) × (d_3 − d_1)| = 1 2 |d_1 ×d_2 + d_2 ×d_3 + d_3 ×d_1| (6)

The total surface S is the sum

P_3 n=0 S_n.

Volume of a Tetrahedron

The volume of a figure bounded by three vectors in space is the triple product of those vectors. Any tetrahedron is 1/6 of such a figure, hence V = 1 6 |d_1 · d_2 × d_3| (7) = 1 6 fi fi fi fi d_{1x} d_{1y} d_{1z} d_{2x} d_{2y} d_{2z} d_{3x} d_{3y} d_{3z} fi fi fi fi (8)

Center of Circumscribed Sphere

To find the circumscribed sphere, we need the point that is equidistant from all four vertices, i.e. we want r such that

(r − r_n) · (r − r_n) = ρ^2; ∀n = 0, 3 r^2 − 2r · r_n + r^2_n = ρ^2

If we take point 0 as the origin, that is, if we use the d_n vectors in place of the r_n, then r^2 = ρ^2, the sphere radius, and the above 4 equations reduce to

2r · d_n = d^2_n ; n = 1, 3

Table 2. Values for regular tetrahedron Quantity Value S_0 = √ 3/4 S = √ 3 V = √ 2/12 ρ = √ 6/4 V_∘ =^4_3 π

�_3 8 � 3_2

This yields the matrix equation for the center of the sphere

2 0 @ d_{1x} d_{1y} d_{1z} d_{2x} d_{2y} d_{2z} d_{3x} d_{3y} d_{3z}

1 A 0 @ x y z

1 A = 0 B @ d^2_1

d^2_2

d^2_3

1 C A (9)

which can be solved for the vector (x, y, z) and the radius of the sphere ρ^2 = x^2+y^2+z^2. Note that the leftmost matrix in equation 9 is the same as the one whose determinant yields the volume of the tetrahedron (equation 8).

The volume of the circumscribed sphere is then

V_∘ = 4 3 πρ^2 (10)

The Regular Tetrahedron The regular tetrahedron of unit side is the ideal against which the true figure of the four spacecraft is to be measured. We may take

d_0 = (0, 0, 0) d_1 = (1, 0, 0)

d_2 = 1 2 , √ 3 2 , 0 !

d_3 = 1 2 , √ 3 6 , √ 6 3 !

Values for the regular tetrahedron of unit side length are listed in Table 2.

Calculating the Quality Factors The quality factors in equations 1 and 2 can now be found with the help of these formulas.

For Q_G, we average the 6 distances between the 4 points to get the side L of the “ideal” regular tetrahedron, with volume L^3 √ 2/12 and surface L^2 √ 3. The true volume and surface are found from equations 7 and 3–6.

For Q_R, the radius of the circumscribing sphere is calculated from equation 9. The actual volume of the sphere need not be calculated, for all the factors just go into the normalizing N .

Q_R = 9 √ 3 8 V ! 1_3

· ρ^{−1}

A Fortran program at the end of this paper calculates both these parameters.

References Robert, P. and Roux, A. (1993). Influence of the Shape of the Tetrahedron on the Accuracy of the Estimate of the Current Density. Proceedings of ESA ‘START’ Conference, Aussois, France, Centre de Recherches en Physique de l’Environment, Issy les Moulineaux, France.

vom Stein, R., Glassmeier, K.-H., and Dunlop, M. (1992). A Configuration Parameter for the Cluster Satellites. Tech. Rep. 2/1992, Institut für Geophysik und Meteologie der Technischen Universität Braunschweig.

A Fortran Subroutine to Calculate the Parameters

      subroutine TETRAQ(r,qg,qr)
c
c To calculate the Glassmeier (QG) and Robert/Roux (QR) quality
c factors for a tetrahedron.
c
c Application: CLUSTER SCIENCE DATA SYSTEM
c (These are to be two auxiliary parameters)
c
c Input: R(3,4) = positions of the 4 points
c
c Output: QG = Glassmeier factor
c QR = Robert/Roux factor (their factor number 10)
c
c Inputs and outputs are single precision, internal calculations
c are double precision
c
c True volume True surface
c QG = ----------- + ------------ + 1.
c Ideal vol Ideal surf
c
c where the ideals are volume and surface of a regular tetrahedron
c of side length equal to the mean of the 6 sides.
c
c | True volume |(1/3)
c QR = Factor * | ----------- |
c | Sphere vol |
c
c where the sphere is that circumscribing the tetrahedron (all 4
c points on the surface), and the factor is such that QR=1
c for a regular tetrahedron.
c
c*********************************************************************
c Patrick W. Daly daly@linmpi.mpg.de
c Max-Planck-Institut
c fuer Aeronomie
c D-37191 Katlenburg-Lindau
c Germany
c 1994 June 7
c*********************************************************************
      implicit none
      real r(3,4),qg,qr
      double precision d(3,3),c(3,3),s1,s2,s3,s0,s,l1,l2,l3,vol
      double precision v(3),w,smean,vmean,lmean,rc
      double precision dot
      integer n,m,k
c
c Find the differences
c
      do n=1,3
        w=dble(r(n,1))
        do m=1,3

          d(n,m)=dble(r(n,m+1))-w
        enddo
      enddo
c
c Find the average side length of all 6 sides
c
      l1=dsqrt(dot(d(1,1),d(1,1)))
      l2=dsqrt(dot(d(1,2),d(1,2)))
      l3=dsqrt(dot(d(1,3),d(1,3)))
      w= l1 + l2 + l3
      do n=1,3
        v(n)=d(n,2)-d(n,1)
      enddo
      w=w + dsqrt(dot(v,v))
      do n=1,3
        v(n)=d(n,3)-d(n,1)
      enddo
      w=w + dsqrt(dot(v,v))
      do n=1,3
        v(n)=d(n,3)-d(n,2)
      enddo
      w=w + dsqrt(dot(v,v))
      lmean=w/6.d0
c
c Find the cross products
c
      do n=1,3
        m=mod(n,3) + 1
        k=mod(m,3) + 1
        call cross(d(1,m),d(1,k),c(1,n))
      enddo
c
c Find the volume of the tetrahedron
c
      vol=dabs(dot(d(1,1),c(1,1)))/6.d0
c
c Find the area of the 4 surfaces and their sum
c
      s1=0.5d0*dsqrt(dot(c(1,1),c(1,1)))
      s2=0.5d0*dsqrt(dot(c(1,2),c(1,2)))
      s3=0.5d0*dsqrt(dot(c(1,3),c(1,3)))
      do n=1,3
        v(n)=c(n,1) + c(n,2) + c(n,3)
      enddo
      s0=0.5d0*dsqrt(dot(v,v))
      s=s0 + s1 + s2 + s3
c
c Find the volume and total area for reg tetrahedron with mean side
c
      vmean=dsqrt(2.d0)*lmean*lmean*lmean/1.2d1
      smean=dsqrt(3.d0)*lmean*lmean
c
c Calculate the Glassmeier factor
c
      w= vol/vmean + s/smean + 1.d0
      qg=sngl(w)
c
c Find the center of the circumscribed circle
c
      w=1.d0/(1.2d1*vol)
      v(1)=w*(c(1,1)*l1*l1 + c(1,2)*l2*l2 + c(1,3)*l3*l3)
      v(2)=w*(c(2,1)*l1*l1 + c(2,2)*l2*l2 + c(2,3)*l3*l3)
      v(3)=w*(c(3,1)*l1*l1 + c(3,2)*l2*l2 + c(3,3)*l3*l3)
      rc=dsqrt(dot(v,v))
c
c Calculate the Robert/Roux factor
c
      w=9.d0*dsqrt(3.d0)/8.d0
      w=(w*vol)**(1.d0/3.d0)/rc
      qr=sngl(w)
c
      return
      end

      function DOT(v,w)
c
c To return the dot product of vectors V and W
c
      implicit none
      double precision dot,v(3),w(3)
      dot=v(1)*w(1) + v(2)*w(2) + v(3)*w(3)
      return
      end

      subroutine CROSS(v,w,x)
c
c To return X as the cross product of vectors V and W
c
      implicit none
      double precision x(3),w(3),v(3)
      x(1)=v(2)*w(3) - v(3)*w(2)
      x(2)=v(3)*w(1) - v(1)*w(3)
      x(3)=v(1)*w(2) - v(2)*w(1)
      return
      end